

# Nonautonomous systems

Marc R. Roussel

November 1, 2005

## 1 Introduction

We have so far focused our attention on autonomous systems. Nonautonomous systems are of course also of great interest, since systems subjected to external inputs, including of course periodic inputs, are very common.

In order to motivate our treatment of nonautonomous systems, we begin with some very simple examples of systems whose state spaces consist of a single variable. As we go, we will find that nonautonomous systems are in some ways different from autonomous systems, and in other ways alike.

**Example 1.1** Consider the very simple system

$$\frac{dx}{dt} = \sin t, \tag{1}$$

with initial condition  $x(0) = x_0$ . First of all, note that this system has no true equilibrium points:  $dx/dt = 0$  when  $\sin t = 0$ , which happens when  $t = n\pi$ , for any integer  $n$ . However, since  $t$  never stops changing,  $dx/dt$  is never zero for more than an instant. It is of course also possible to make up autonomous systems which lack equilibria (e.g.  $\dot{x} = 1$ ), but these often have uninteresting behavior.

The state space  $x$  is no longer a proper phase space for nonautonomous differential equations because the behavior at a given point in the state space depends on the time at which that point was reached. Essentially, this means that the phase space is two-dimensional and consists of the variables  $x$  and  $t$ . We can in fact formalize this by expanding our original equation to

$$\frac{dx}{d\tau} = \sin t \tag{2a}$$

$$\text{and } \frac{dt}{d\tau} = 1. \tag{2b}$$

In practical terms, we haven't gained much. Conceptually however this transformation highlights the fact that a nonautonomous system with a  $d$ -dimensional state space is in a sense




Figure 1: Phase space for equation 1. The solution is shown as a red curve. Because only the value of  $t$  modulo  $2\pi$  is needed, the simplest representation of the phase space is as a cylinder.

equivalent to a  $d + 1$ -dimensional autonomous system. The evolution of the time variable is of course trivial, so a nonautonomous system isn't equivalent to a  $d + 1$ -dimensional autonomous system of general form. Accordingly, Hale and Koçak, in their book *Dynamics and Bifurcations*, call these systems “ $d$  and a half-dimensional” (e.g.  $1\frac{1}{2}$ -dimensional for the current example). There is some merit to this nomenclature, although I prefer to simply say that I'm working with a nonautonomous system.

Nonautonomous terms don't have to be periodic, but in applications, they often are. That being the case, there is some additional structure to the equations we can exploit. The solution of our differential equation is

$$x(t) = x_0 + 1 - \cos t. \quad (3)$$

We can draw these solutions in our  $(t, x)$  phase space. They are of course just sine waves. Of course, the behavior repeats whenever  $t$  has increased by  $2\pi$ . We therefore only need to know  $t$  modulo  $2\pi$ , i.e. the remainder after subtracting out any integer multiples of  $2\pi$  from  $t$ . This means that we could think of our phase space not as a plane but as a cylinder, where  $t$  goes from 0 to  $2\pi$  and then repeats again. Figure 1 illustrates the phase space for this example.

We can make good use of the fact that phase space is a cylinder for periodically driven systems. Suppose that we take a line of constant  $t \pmod{2\pi}$ . Trajectories intersect this line once each driving period. For equation 1, there is a single intersection given by equation 3. We can conclude that the orbit is periodic. Even if we weren't able to write down the solution, we would still know the topology of the phase space, and be able to study the intersections numerically. We would notice that we always hit the line  $t \pmod{2\pi} = t_s$  at the same point, for some given  $t_s$ , and again conclude that the orbit is a simple closed curve. We could also vary the initial conditions, noticing that we get a different intersection for each different value of  $x_0$ . We would therefore conclude that the periodic solution is *not* a limit

cycle, but is in fact some form of conservative oscillation. It turns out that system 2 is in fact Hamiltonian.

Describing the motion in terms of the intersections of the solution with a given line (or surface, in general) is an important technique of nonlinear dynamics. The line  $t \bmod 2\pi = t_s$  is called a **surface of section**, or **Poincaré section**. We will return to this idea shortly.

**Example 1.2** Consider the differential equation

$$\frac{dx}{dt} = -x + \cos t,$$

with initial condition  $x = x_0$  at  $t = 0$ . If you happen to know the variation of constants formula, you can solve this equation by hand, or you can just use Maple:

```
> dsolve({diff(x(t),t)=-x(t)+cos(t),x(0)=x0},x(t));
```

$$x(t) = \frac{1}{2} \cos(t) + \frac{1}{2} \sin(t) + e^{(-t)} \left(-\frac{1}{2} + x_0\right)$$

Note that at long times, all solutions approach the unique solution

$$x(t) = \frac{1}{2} (\cos t + \sin t).$$

This is a limit cycle in the  $x \times (t \bmod 2\pi)$  phase space: It's a solution of the governing equation, and the general solution clearly implies that the system returns to this solution if it is displaced from it slightly.

The two foregoing examples hint that most of the behaviors which are possible in a planar dynamical system are also possible in a nonautonomous system. As mentioned above, the one thing we can't have is a true equilibrium point, unless of course we extend our notion of equilibrium.

**Example 1.3** Consider the differential equation

$$\frac{dx}{dt} = -x + \frac{1}{t} - \frac{1}{t^2}.$$

As  $t \rightarrow \infty$ ,  $dx/dt \rightarrow -x$ . It follows that we eventually get solutions which exponentially decay to zero. Since the nonautonomous term is not periodic, the phase space is now the entire  $(t, x)$  plane. In a sense, this system therefore has an equilibrium point at  $(+\infty, 0)$  which is reached for initial conditions with  $t > 0$ .

It is often convenient to describe equilibrium points at infinity. In some cases, we can work with them directly as we have done here. In others, it is easier to transform to a coordinate

system which places these equilibria at finite coordinates. In this case, we are interested in the planar system

$$\begin{aligned}\frac{dx}{d\tau} &= -x + \frac{1}{t} - \frac{1}{t^2}, \\ \frac{dt}{d\tau} &= 1.\end{aligned}$$

We want to change our coordinate system such that the interval  $[0, \infty)$  is mapped into some finite interval, say  $[0, 1]$ . There are many possible transformations, of which the following is particularly convenient:

$$\theta = \frac{t}{t+1}.$$

By the chain rule,

$$\begin{aligned}\frac{d\theta}{d\tau} &= \frac{d\theta}{dt} \frac{dt}{d\tau} \\ &= \frac{1}{(t+1)^2}.\end{aligned}$$

$$\text{Also, } t = \frac{\theta}{1-\theta}.$$


$$\therefore \frac{d\theta}{d\tau} = (1-\theta)^2$$

$$\text{and } \frac{dx}{d\tau} = -x + \frac{1-\theta}{\theta} - \left(\frac{1-\theta}{\theta}\right)^2.$$

The point  $(\theta^*, x^*) = (1, 0)$  is clearly an equilibrium point of the transformed system. The system is now in a form in which we can study the stability of this point, and by inference the stability of the equilibrium point at infinity of the original system.

## 2 Poincaré sections

Recall the Brusselator model:



After reducing to dimensionless form, we had obtained the equations

$$\begin{aligned}\dot{x} &= a - bx + x^2y - x, \\ \dot{y} &= bx - x^2y.\end{aligned}$$

In our earlier work with this model, it was assumed that  $a$  and  $b$  were constants. Now suppose instead that the chemical A is supplied periodically:

$$a = a_0 + \alpha \sin(2\pi t/T), \quad (4)$$

where  $T$  is the period of the input, and  $a_0$  and  $\alpha$  are constants. We can't solve these equations analytically. However, we can use some of the ideas developed above to study this system.

First, since the input is periodic, we work in the phase space  $a \times b \times (t \bmod T)$ . A periodic phase space of this sort is available in xpp. We use the following input file:

```
# Brusselator ode file

# Differential equations:
x' = a0 + alpha*sin(2*pi*theta/period) - b*x + x^2*y - x
y' = b*x - x^2*y
theta' = 1

# Initial conditions
x(0) = 0.9
y(0) = 0.9

# Default values of the parameters:
param a0=1, b=3
param alpha=0.01, period=1

# Reserve lots of storage space
# in case we want to get a long trajectory
@ MAXSTOR=1000000

done
```

Because xpp always uses  $t$  for time, I've had to change our naming convention, using  $\theta$  for the time in our autonomous set. After starting xpp, click on `phAsespace`→`Choose`. Set the period to 1, then select  $\theta$  in the dialog box which appears and click on `Done`. Click on `Viewaxes` and arrange for  $x$  to be plotted vs  $\theta$ . Finally, set `Total` to 200 and `tRansient` to 100 in the `Numerics` menu. This will give us a reasonably long trajectory while discarding any short-lived transient behavior. If you run the model, and then click on `Window/zoom`→`Fit`, you will see a somewhat complicated picture (Fig. 2). Remember that we're looking at trajectories on a cylinder.<sup>1</sup> The right and left edges of the image are therefore associated. What is happening effectively is that the period of the system for these parameters is different from the driving period. We can confirm this by looking at (e.g.)  $x$  vs  $t$ , which clearly shows that the system oscillates with a period of about 7 time units. We have two irrationally related frequencies, so we have a **quasiperiodic orbit** on the cylinder. This orbit will in time completely cover the cylinder.

We get a quasiperiodic orbit in this case because the driving is too weak and too far removed from the system's natural frequency to entrain it. Change  $\alpha$  to 0.5 and  $\text{period}$  to 5 in the xpp parameter window. Don't forget to also change the period in the `phAsespace` dialog to 5. The

---

<sup>1</sup>The geometry is a little more complicated than that because there's a third variable,  $y$ , but the image of a cylinder is a useful one.




Figure 2: A trajectory of the driven Brusselator projected onto the  $x \times (t \bmod 2\pi)$  plane for  $a_0 = 1$ ,  $b = 3$ ,  $\alpha = 0.01$  and  $T = 1$ .

solution is now a limit cycle (Fig. 3), as you can verify by doing an Initialconds→Last, then manually perturbing the initial conditions, and running the model again. The solution curves with transients discarded lie right on top of each other.

Pictures like that shown in Fig. 2 are suggestive, but how do we know that the trajectory really is quasiperiodic rather than chaotic? These two types of nonrepeating behavior aren't always easy to tell apart. We could of course start a second trajectory with similar initial conditions to the first. If the system is chaotic, a plot of  $x$  vs  $t$  should be noticeably different for the two trajectories. On the other hand, a quasiperiodic solution would just exhibit a small phase shift. Again though, these two possibilities aren't always as easy to tell apart as we would like.

The solution to this conundrum is the Poincaré section introduced in example 1.1. In this case, the Poincaré section is two-dimensional, consisting of  $(x, y)$  pairs. We're going to collect and plot values of  $x$  and  $y$  every time  $t \bmod T = t_s$ . The following are possible outcomes of this numerical experiment:

- The Poincaré section might contain a single point, indicating a closed orbit, in this case a limit cycle.
- The points in the Poincaré section might draw out a closed curve. This would indicate that the variables are cycling at an additional frequency not equal to the driving frequency, i.e. that the behavior is quasiperiodic.
- A chaotic trajectory would show up as a complex distribution of points in the section.




Figure 3: A trajectory of the driven Brusselator projected onto the  $x \times (t \bmod 2\pi)$  plane for  $a_0 = 1$ ,  $b = 3$ ,  $\alpha = 0.5$  and  $T = 5$ .

Suppose that we want to study the trajectory shown in Fig. 2 by this technique. We start by putting the parameters and the phase space period back to their original values. Then click on nUmeric→Poincare map→Section. A dialog box will appear. Set the variable to theta, the section to 0.5, the direction to 1, and Stop on sect to N. This will cause xpp to store values of the variables whenever *theta* increases through 0.5. (The latter value is, of course, purely arbitrary.) If you now run the model and plot *y* vs *x* with dots (Graphic stuff→Edit curve 0, and set the Line type to zero), you should see something like Fig. 4. If you continue the trajectory for a longer time, the curve fills in. This is a clear indication of quasiperiodic behavior.

Fig. 5 shows a set of results for  $a_0 = 0.4$ ,  $b = 1.2$ ,  $\alpha = 0.05$  and  $T = 7.8$ . Note that the figure on the Poincaré section (panel a) is no longer a closed curve. It isn't even continuous! Similarly, the cylinder plot of the orbit (panel b) isn't space filling. In panel (c), we see what looks like a strange attractor. All of these pieces of evidence taken together suggest chaos. The best way to verify is of course to look for sensitive dependence on initial conditions. This is what we do in panel (d). Because of the periodic input, the maxima and minima of the two solutions occur at roughly the same times, but their amplitudes rapidly become decorrelated.

### 3 Automated bifurcation analysis

Having done a little bit of preliminary exploration, we might want to use AUTO to find the bifurcation curves. AUTO is very picky about the types of systems it can handle. It doesn't handle




Figure 4: Poincaré section for the parameters of Fig. 2. The section was taken at  $t \bmod T = 0.5$ .

nonautonomous systems. It also doesn't know anything about cylindrical phase spaces, and it's not much use for systems in which one variable runs away to infinity, so our trick of adding a differential equation for  $t$  won't be of much help here.

What we need to do instead is to add differential equations whose *solutions* are our desired nonautonomous terms. Because we don't really want to study the stability of these added equations, our nonautonomous terms should furthermore be stable solutions of these equations. For sinusoidal inputs, the following equations have the desired properties:

$$\begin{aligned} \dot{u} &= u(1 - u^2 - v^2) - 2\pi v/T, \\ \dot{v} &= v(1 - u^2 - v^2) + 2\pi u/T. \end{aligned}$$

With initial conditions  $(u(0), v(0)) = (1, 0)$ , the solution of these equations is

$$(u(t), v(t)) = (\cos(2\pi t/T), \sin(2\pi t/T)).$$

Here is an xpp input file for this extended system:

```
# Driven Brusselator ode file

# Differential equations:
x' = a0 + alpha*v - b*x + x^2*y - x
y' = b*x - x^2*y
u' = u*(1-u^2-v^2) - 2*pi*v/period
v' = v*(1-u^2-v^2) + 2*pi*u/period
```






Figure 5: Some numerical integration results for the driven Brusselator with  $a_0 = 0.4$ ,  $b = 1.2$ ,  $\alpha = 0.05$  and  $T = 7.8$ . Panel (a) shows a Poincaré section taken at  $t \bmod T = 1$  integrated for 10 000 time units with a transient of 1500 time units. Panel (b) shows a trajectory on the cylindrical phase space. Panel (c) shows a trajectory in the  $(x, y)$  state space. Both of the latter were from a simulation corresponding to 3000 units of simulation time, again with a 1500 unit transient omitted. Panel (d) shows a pair of trajectories started from nearby points (noted in the legend) at  $t = 0$ .

```

# Initial conditions
x(0) = 0.9
y(0) = 0.9
u(0) = 1
v(0) = 0

# Parameter values we want to study:
param a0=0.4, b=1.2
param alpha=0.5, period=7.8

# Reserve lots of storage space
# in case we want to get a long trajectory
@ MAXSTOR=1000000

done

```

We're going to use this expanded model to study the driven Brusselator in the parameter range of Fig. 5. Note that I've set  $\alpha = 0.5$ , but that all other parameters are set as in the figure.


Because of the periodic driving, we won't be able to identify a stable steady state from which to start AUTO.<sup>2</sup> AUTO can also start from periodic solutions, but it's tricky. For one thing, the whole system has to be expressing the same period. In other words, we can't start from a quasiperiodic solution. That means that we have to find a solution in which the Brusselator locks onto the driving signal and produces a simple limit cycle in our cylindrical phase space. That's why I chose an initial value of 0.5 for the parameter  $\alpha$ , a value which I found using the previous `xpp` input file and its cylindrical phase space. Note that this is an unphysical value for this parameter since  $a$  can then be negative (equation 4). That doesn't matter for our current purposes.

Once we have started `xpp` with our new input file, we need to integrate long enough to get rid of any transients. We therefore run for 2000 units of simulation time, then hit `Initialconds`→`Last`. This gives us some initial conditions on the limit cycle. When studying periodic solutions, AUTO needs an estimate of the period to be stored in the `Total` integration time. Because we're starting with a simple limit cycle, we know exactly what the period is: It's 7.8 units. If you right-click in the plot area, `xpp` will print out the  $x$  and  $y$  coordinates on the screen (not to be confused with the  $x$  and  $y$  variables in this model). You can therefore read off the times corresponding to two successive maxima or minima and verify that the period is indeed 7.8. If you set `Numerics`→`Total` to this value, and then do `Initialconds`→`Last` a few times, you will find that the curves plot right on top of each other, which confirms that the period is indeed 7.8.

It's time to start AUTO (`File`→`Auto`). Click on `Axes`→`Hi` and fill in the values as follows:

---

<sup>2</sup>We could fix this by (a) modifying the equations of the  $(u, v)$  system such that they have a stable steady state for some value of a newly introduced parameter, (b) setting  $b$  below the Andronov-Hopf bifurcation, and perhaps (c) setting  $\alpha = 0$ . We could then follow solution branches within AUTO to get to our desired initial parameters. In this case though, this seems like rather a lot of work, and there is an alternative.



In the Numerics dialog, set  $D_s$  to  $-0.01$ ,  $D_{smin}$  to  $0.0001$ , and  $D_{smax}$  to  $0.1$ . Also set  $Par\ Max$  to  $0.5$ . We want to follow a branch of periodic solutions, so click on  $Run \rightarrow Periodic$ . AUTO finds a period-doubling bifurcation at  $\alpha = 0.475$ . If we grab this point and ask AUTO to follow the period-doubling branch, it finds the following additional interesting points:

- a period-doubling point at  $\alpha = 0.145$ ,
- another period-doubling point at  $\alpha = 3.32 \times 10^{-2}$ , and
- a torus bifurcation at  $\alpha = 8.46 \times 10^{-3}$ .

(It actually finds each of these points over and over again. The small differences between the values of  $\alpha$  from one pass to the other gives you an idea of how accurate these computations are. Once you notice this happening, you can hit the ABORT button to stop AUTO.)

If we continue to grab and run continuations on the period-doubling branches, we find the following period-doubling points:

$\alpha$	Period
$4.75 \times 10^{-1}$	7.8
$1.45 \times 10^{-1}$	15.6
$9.83 \times 10^{-2}$	31.2
$9.05 \times 10^{-2}$	62.4
$4.24 \times 10^{-2}$	31.2
$3.32 \times 10^{-2}$	15.6

Note that the period doubling points accumulate near  $\alpha = 0.09$ . Presumably, there is a similar accumulation at the low end, perhaps around  $\alpha = 0.05$ , although I couldn't find as many period-doubling points there as at the higher end. If I tried a bit harder (smaller values of  $D_s$  and  $D_{smax}$ ), I might find a few more period-doubling points. In any event, it's pretty clear that the chaotic region is somewhere around values of  $\alpha$  between  $0.05$  and  $0.09$ , which is where we found it earlier (Fig. 5).

There's still the matter of exploring what the torus bifurcation does, so let's go back to xpp. Again, our first xpp input file is most convenient for this kind of thing. This bifurcation occurs for  $\alpha = 8.46 \times 10^{-3}$ , so let's take a couple of values of  $\alpha$  to either side of this value, say  $8.4 \times 10^{-3}$  and  $8.5 \times 10^{-3}$ . This turns out to be trickier than might first appear, so here's the procedure I followed:




Figure 6: (a) Limit cycles obtained for  $a_0 = 0.4$ ,  $b = 1.2$ ,  $T = 7.8$  and various values of  $\alpha$  (distinguished by color). (b) Quasiperiodic solution at  $\alpha = 0.0084$ . (c) Quasiperiodic solution at  $\alpha = 0.009$ .

1. Start at  $\alpha = 0.01$  and compute the trajectory with a very long transient (5000 time units). We get a closed orbit (Fig. 6a).
2. Decrease  $\alpha$  to 0.009, and do an `Initialconds`→`Last`. This starts the trajectory from the periodic orbit found for  $\alpha = 0.01$ , which we may hope is close to the trajectory for  $\alpha = 0.009$ . We again get a closed orbit (Fig. 6a).
3. Decrease  $\alpha$  to 0.0085, and do `Initialconds`→`Last` again. This time, we get something slightly messy, so do `Initialconds`→`Last` a few more times. The problem we're encountering is called **critical slowing down**, which just means that the transients are longer close to bifurcation points. We eventually recover a periodic orbit (Fig. 6a).
4. Repeat the above procedure for  $\alpha = 0.0084$ . This time, we get a complex (quasiperiodic) orbit, no matter how many times we do `Initialconds`→`Last` (Fig. 6b).

The torus bifurcation has converted our limit cycle into a quasiperiodic solution. We can picture quasiperiodic solutions as windings around a torus. Physically, we have simply dropped the value of  $\alpha$  to such a low value that the periodic term is no longer strong enough to entrain the Brusselator, so we see both the Brusselator's natural frequency and the driving signal's frequency in the solutions. Mathematically, you can picture this as a bifurcation of a closed curve into a torus.

The reason that we had to be so careful in going through this exercise is that, over a narrow

range of parameters, there is **bistability** between the torus and the limit cycle. To see this, increase  $\alpha$  back to 0.009 and do an `Initialconds`→`Last` from the solution computed at the smaller value of  $\alpha$ . A quasiperiodic solution appears (Fig. 6c). Which attractor we reach depends on the initial conditions. The lesson here is that, when doing this kind of work, it's never a bad idea to try approaching the bifurcation point from both directions. You often discover interesting things that way.